Math 2270-1 Fall 2005
Solutions to Part A of first Maple project

This is a group project, up to 3 people may work in a single group. All group members will receive the same group grade.

Part A: some matrix algebra questions

(These questions are modified from problems on page 27 of the text Multivariable Mathematics with Maple, by J.A. Carlson and J.M Johnson.) You are to create a document in which you answer the following questions, via a mixture of Maple computations and textual insertions. You are to print out a copy of this document to hand in, as part A of your first Maple project. Don't forget to put your name and section number on it!

Define, using Maple's "old" linear algebra package "linalg",

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

$$B := \begin{bmatrix} 2 & 1 & 0 \\ 1 & 2 & 1 \\ 0 & 1 & 2 \end{bmatrix}$$

```
> with(linalg):
> A:=matrix(3,3,[1,2,3,4,5,6,7,8,9]);
  B:=matrix(3,3,[2,1,0,1,2,1,0,1,2]);
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

$$B := \begin{bmatrix} 2 & 1 & 0 \\ 1 & 2 & 1 \\ 0 & 1 & 2 \end{bmatrix}$$

```
>
```

1a) Compute AB and BA. Are they the same?

```
> AB:=evalm(A&*B);BA:=evalm(B&*A);
  #matrix muliplication does not usually commute!
```

$$AB := \begin{bmatrix} 4 & 8 & 8 \\ 13 & 20 & 17 \\ 22 & 32 & 26 \end{bmatrix}$$

$$BA := \begin{bmatrix} 6 & 9 & 12 \\ 16 & 20 & 24 \\ 18 & 21 & 24 \end{bmatrix}$$

We see that AB does not equal BA.

1b) Compute A+B and B+A. Are they the same?

```
> evalm(A+B)=evalm(B+A); #matrix addition is commutative
```

$$\begin{bmatrix} 3 & 3 & 3 \\ 5 & 7 & 7 \\ 7 & 9 & 11 \end{bmatrix} = \begin{bmatrix} 3 & 3 & 3 \\ 5 & 7 & 7 \\ 7 & 9 & 11 \end{bmatrix}$$

```
>
```

1c) Define C to be A+B. Compute C^2 and compare it to A^2 + 2AB + B^2. Are they the same? Can you think of a small change you could make in the expression ''A^2 + 2AB + B^2'' in order to make it equal to C^2? Justify your answers!

```
> C:=A+B;
  evalm(C^2);
```

$$C := A + B$$

$$\begin{bmatrix} 45 & 57 & 63 \\ 99 & 127 & 141 \\ 143 & 183 & 205 \end{bmatrix}$$

```
> evalm(A^2+2*A&*B+B^2);
```

$$\begin{bmatrix} 43 & 56 & 59 \\ 96 & 127 & 134 \\ 147 & 194 & 207 \end{bmatrix}$$

```
>
```

We see that C^2 does not equal A^2 + 2AB+B^2. If we use the distributive property of multiplication over addition correctly, and note that AB does not equal BA, we see that

$$[A+B]^2 = A\,[A+B] + B\,[A+B]$$
$$= A^2 + AB + BA + B^2$$

```
> evalm(A^2+A&*B+B&*A+B^2); #this will equal C^2!
```

$$\begin{bmatrix} 45 & 57 & 63 \\ 99 & 127 & 141 \\ 143 & 183 & 205 \end{bmatrix}$$

```
>
```

1d) Define v=(1,2,3) to be a vector. Compute Av. What does Maple give you when you try vA? Explain.

```
> v:=vector([1,2,3]);
```

$$v := [1, 2, 3]$$

```
> evalm(A&*v);evalm(v&*A);
```

$$[14, 32, 50]$$

$$[30, 36, 42]$$

Apparently Maple will treat a vector as either a row or column vector, as needed so that the multiplication makes sense.

1e) Solve Bx=v for x, where v is the vector in (1d). Get your solution in each of the following three ways: use the rref command on the augmented matrix; use the command ''linsolve''; use the inverse matrix to B.

```
> rref(augment(B,v));
  evalm(inverse(B)&*v);
  linsolve(B,v);
```

$$\begin{bmatrix} 1 & 0 & 0 & \frac{1}{2} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & \frac{3}{2} \end{bmatrix}$$

$$\left[\frac{1}{2}, 0, \frac{3}{2}\right]$$

$$\left[\frac{1}{2}, 0, \frac{3}{2}\right]$$

```
>
```

So the solution is

$$x = \begin{bmatrix} \frac{1}{2} \\ 0 \\ \frac{3}{2} \end{bmatrix}$$

2a) Solve Ax=v for x, where A and v are as indicated above. Verify, with Maple, that your solution x actually solves the equation Ax=v.

```
> inverse(A);
Error, (in inverse) singular matrix
```

A is not invertible!

```
> rref(augment(A,v));
  linsolve(A,v);
```

$$\begin{bmatrix} 1 & 0 & -1 & \frac{-1}{3} \\ 0 & 1 & 2 & \frac{2}{3} \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

$$\left[-\frac{1}{3}+\_t_1, \frac{2}{3}-2\,\_t_1, \_t_1\right]$$

```
>
```

We see that Maple has backsolved the augmented matrix just as we would!

```
> evalm(A&*vector([-1/3+t,2/3-2*t,t]));
  #shows that our solutions solve the system
```

$$[1, 2, 3]$$

```
>
```

2b) Repeat your work above in order to solve Ax=w, where w=(-1,4,1). Explain your answer.

```
> w:=vector([-1,4,1]);
```

$$w := [-1, 4, 1]$$

```
> rref(augment(A,w));
  linsolve(A,w);
```

$$\begin{bmatrix} 1 & 0 & -1 & 0 \\ 0 & 1 & 2 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
>
```

System was inconsistent - as we see from rref of the augmented matrix. The linsolve command returned nothing, to reflect the fact that there was no solution.